/*
	THIS FILES CONSTRUCTS FIGURE 1
*/

*LOAD
use temp\sample, clear
egen f=group(Company_CVR)

*REGRESSIONS OF "INITIAL LETTER GROUP" INDICATOR ON APPROACH INDICATOR
levelsof group, local(levels) clean
local n=1
foreach l of local levels {

	gen temp=(group=="`l'")
	eststo est_`n' : reg temp tdcinitiative , vce(cl f)
	drop temp
	local ++n

}

*CONSTRUCT FIGURE 1
levelsof group, local(levels) clean
local N : word count `levels'
local n=1
local coefplot
foreach l of local levels {
	local add "est_`n', aseq(`l')"
	if `n'<`N' local add "`add'  \ "
	local coefplot "`coefplot' `add'"
	local ++n
}

coefplot (`coefplot'), vertical drop(_cons) swapnames yline(0) xtitle("Firms' initial letter")
graph export output\figure_1.png, replace